



# 오빠두엑셀 기초강좌 E-BOOK

1

Free Tutorial

2

Business Excel

3

Basic to Expert

한국어

함수마스터 1-2

Dynamic

Range

동적범위

INDEX 편



## 1. Dynamic Range (동적범위)를 적용하는 방법

동적범위를 적용하는데 크게 3 가지 방법이 있습니다.

- OFFSET 을 이용한 Dynamic Range**  
 보편적으로 사용되는 동적범위입니다. 속도저하가 발생할 수 있으므로, 작은양의 데이터에 사용하기 적합합니다.  
 자세한 내용은 'OFFSET 동적범위' 강의를 참조해주세요.
- INDEX 를 이용한 Dynamic Range**  
 데이터의 양이 많을 경우, INDEX 를 이용하여 동적범위를 설정하면 보다 빠른 처리속도를 보여줍니다. OFFSET 과 마찬가지로 보편적으로 사용되는 동적범위 수식입니다.
- Table(표기능)을 이용한 Dynamic Range**  
 엑셀에 기본적으로 내장된 표기능을 이용한 동적범위입니다. 초보자도 이용할 수 있는 가장 쉬운 방법이며 다양한 기능을 같이 제공합니다. 하지만 표로 지정된 데이터범위를 유연하게 사용하는데 제한이 있습니다.  
 자세한내용은 '표(TABLE) 활용기초' 강의를 참조해주세요.

장단점비교	OFFSET/INDEX 동적범위	표(TABLE)기능 동적범위
<b>장점</b>	1. 더욱 융통성있고 유연하게 원시자료(Raw Data)를 편집할 수 있음. 2. 표기능보다 적은 단계로 적용가능 3. 워크시트를 더욱 깔끔하게 사용가능	1. 엑셀에 내장된 기능으로 초보자도 사용가능 2. 같은 행의 특정 열에 빈칸이 있더라도 동일한 표로 인식하여 범위를 자동으로 잡아줌. 3. 많은 양의 동적범위가 필요할 경우 작업시간을 크게 줄여줌.
<b>단점</b>	1. OFFSET 과 INDEX 함수에 대한 기본 지식 필요. 2. 해당 동적범위에 '빈칸'이 있을 경우, (특히 데이터유효성-목록에 적용시), 동적범위 안에 낭비가 발생.	1. 표기능의 특성상 연속된 열(좌우로 연속된 범위)에 한하여 동일한 표로 간주됨. 2. '정규화'되지 않은 자료에는 표기능으로 동적범위를 적용하기 어려움.

## 2. INDEX 함수 기초

= INDEX ( 참조범위 , 행번호(위아래) , [ 열번호(좌우) ] )

- ⚠ 행번호와 열번호는 '1'부터 시작  
'0' 입력 시, 전체행/전체열 참조 (참조형 INDEX 함수) ⇒ 결과값: 배열(범위)

	A	B	C	D	E	
1	1	2	3	4		예제 1) =INDEX(\$A\$1:\$D\$4,1,3) ⇒ 3
2	5	6	7	8		=INDEX(\$A\$1:\$D\$4,4,5) ⇒ #N/A
3	9	10	11	12		예제 2) =INDEX(\$A\$1:\$D\$4,0,3) ⇒ {3,7,11,15}
4	13	14	15	16		=SUM(INDEX(\$A\$1:\$D\$4,1,0)) ⇒ 10
5						=SUM(INDEX(\$A\$1:\$D\$4,0,0)) ⇒ 전체합
6						

## 3. INDEX 를 이용한 Dynamic Range (동적범위) 만들기

= 시작지점 : INDEX ( 참조범위 , COUNTA(상하범위) , COUNTA(좌우범위) )

- ⚠ 상하범위: 고유값이 있는 범위 (빈칸 X)  
좌우범위: 머리글 범위 (빈칸 X)  
COUNTA 로 인한 빈칸인식 문제 해결은 '오빠두엑셀 함수마스터 1-3' 강의를 참조하세요.

### 3-A. INDEX 동적범위 예제 1

	A	B	C	D
1	아이디	직원명	성별	
2	1	김진수	남	
3	2	이진아	여	
4	3	박철희	남	
5	4	정희수	남	
6	5	이은지	여	
7	6	김희철		
8	7		여	
9	8	김하나	여	
10	9	이정수	남	
11				
12				
13				
14				

= \$A\$2 : INDEX ( \$A:\$C ,  
COUNTA ( \$A:\$A ) ,  
COUNTA ( \$1:\$1 ) )

분석 1)

= INDEX ( \$A:\$C , COUNTA(\$A:\$A) , COUNTA(\$1:\$1) ) ⇒ 결과값 : \$C\$10

분석 2)

= \$A\$2 : INDEX ( \$A:\$C , COUNTA(\$A:\$A) , COUNTA(\$1:\$1) ) ⇒ 결과값 : \$A\$2:\$C\$10

**3-B. INDEX 동적범위 예제 2**

INDEX 함수를 이용해 한번에 모든 범위를 Dynamic Range(동적범위)로 설정하는 방법

⚠ 단, 해당 시트안에 1 개의 데이터표만 존재해야 함.

	A	B	C	D	E	F	G
1	주문자명	배송주소	판매물건	단가	수량	가격	배송상태
2	김진수	서울 강남	침대	100,000	1	100,000	배송완료
3	이진아	경기도 인천	배게	30,000	2	60,000	배송완료
4	박철희	서울 역삼	이불	60,000	1	60,000	결제대기
5	정희수	경기도 수원	이불	60,000	1	60,000	배송중
6	이은지	경기도 인천	침대	100,000	1	100,000	배송완료
7	김희철	경기도 인천	침대	100,000	1	100,000	배송중
8	김하나	서울 신림	이불	60,000	2	120,000	결제대기
9	이정수	서울 신촌	배게	30,000	2	60,000	배송완료
10							
11							
12							


**= \$A\$1 : INDEX ( \$1:\$104576 , COUNTA ( \$A:\$A ) , COUNTA ( \$1:\$1 ) )**

⚠ 엑셀 2003 이전버전은, 65536 개의 행만 지원


**4. 자동 업데이트 차트 만들기**

**4-A. 표(Table) 기능 이용하여 자동 업데이트 차트 만들기**


1. 자동 업데이트 차트 (표) 시트로 이동하여 데이터 아무 곳이나 클릭한 뒤 [삽입] - [표] 를 클릭하여 표 만들기를 통해 선택된 범위를 표로 변환합니다. (단축키 Ctrl + T)





2. 변환된 표를 선택한 뒤, [삽입] - [추천차트]를 클릭하여 차트를 삽입합니다.




3. 매출액과 판매개수 사이에 격차가 크므로, 그래프를 좀 더 보기 좋게 변경하기 위한 작업을 합니다.



4. 차트에서 우클릭하여 [차트종류변경] - [콤보]를 선택한 뒤 [판매개수]의 '보조축'에 체크합니다



5. 이후 표에 내용을 업데이트 하면 삽입된 차트가 자동으로 갱신됩니다.




- ⚠ 표기능을 이용할 경우, 아래 두가지 사항을 주의해주세요.
  - 병합된 셀을 표로 변환할 경우, 병합된 셀이 분리됩니다.
  - 변환된 표는 연속된 행/열의 데이터만 동일한 표의 내용으로 인식합니다.

#### 4-B. INDEX 동적범위 응용하여 자동 업데이트 차트 만들기

INDEX 함수를 이용한 동적범위 뿐만 아니라, OFFSET 함수를 이용한 동적범위도 자동업데이트 차트를 만들기 위한 동적범위로 똑같이 적용 가능합니다.


1. [수식] - [이름관리자]로 이동하여 새로운 사용자지정범위를 아래와 같이 생성합니다.




rng 날짜	=A\$2:INDEX(\$A:\$A,COUNTA(\$A:\$A))
rng 매출액	=B\$2:INDEX(\$B:\$B,COUNTA(\$B:\$B))
rng 판매개수	=D\$2:INDEX(\$D:\$D,COUNTA(\$D:\$D))

- ⚠ 차트에 들어가는 데이터는 '머릿글을 제외한' 범위입니다.
  - 셀병합이 된 범위를 동적범위로 설정할 시, 실제 데이터가 들어가는 범위로 설정하는 것에 주의하세요. (예: [B:C]의 병합된 셀의 경우, 실제 데이터는 [B]셀에 입력됨)


2. 차트로 변환할 범위를 선택한 뒤, [삽입] - [추천차트]로 이동하여 차트를 삽입하세요.




3. 매출액과 판매개수 사이에 격차가 크므로, 그래프를 좀 더 보기 좋게 변경하기 위해 차트에서 우클릭하여 [차트종류변경] - [콤보]를 선택한 뒤 [판매개수]의 '보조축'에 체크합니다.



4. 차트에 연동된 데이터계열을 방금 생성한 '동적범위'로 바꾸기 위해, 차트에서 우클릭한 뒤 [데이터선택]을 클릭합니다.




5. 병합된 셀로 인해 나타난 <빈데이터계열>은 제거한 뒤, [범례항목]에서 '매출액'과 '판매개수'의 [편집]을 클릭하여 동적범위로 계열값을 변경해줍니다.




⚠ 계열값을 동적범위로 설정할 때, 입력된 범위의 워크북을 지정해주는 느낌표(!) 뒤에 동적범위를 입력하는 것에 주의하세요. (예: '시트명'!rng 범위)

6. 가로축항목(레이블)에서도 [편집]을 클릭하여 동적범위로 계열값을 변경해줍니다.



7. 표 안에 새로운 데이터를 추가하면 차트가 자동으로 업데이트 되는 것을 확인 할 수 있습니다.



자세한 내용은 오빠두엑셀 함수마스터 1-2 영상강의를 참조하세요!